

<programa> ::= <comando>

<comando> ::= <if-else> | <atribuicao> | "{" <lista-comandos> "}" | ε

<lista-comandos> ::= <comando> { <comando> }

<if-else> ::= "if" "(" <expressao-logica> ")" <comando> ["else" <comando>]

<atribuicao> ::= <identificador> "=" <expressao-arit> ";"

<expressao-logica> ::= <expressao-logica> "||" <expressao-and>

| <expressao-and>

<expressao-and> ::= <expressao-and> "&&" <expressao-comp>

| <expressao-comp>

<expressao-comp> ::= <expressao-arit> <op-rel> <expressao-arit>

| "!" <expressao-comp>

| "(" <expressao-logica> ")"

| <expressao-arit> (* trata número ≠ 0 como verdadeiro *)

<op-rel> ::= "==" | "!=" | "<" | "<=" | ">" | ">="

<expressao-arit> ::= <expressao-arit> "+" <termo>

| <expressao-arit> "-" <termo>

| <termo>

<termo> ::= <termo> "*" <fator>

| <termo> "/" <fator>

| <fator>

<fator> ::= "-" <fator>

| <constante>

| <identificador>

| "(" <expressao-arit> ")"

<constante> ::= <numero>

<numero> ::= <digito> {<digito>} ["." <digito> {<digito>}]

<identificador> ::= <letra> { <letra> | <digito> | "_" }

<letra> ::= "a".. "z" | "A".. "Z"

<digito> ::= "0".. "9"

Exemplo de Entrada Aceita:

```
if (a + 2*b > 3 && c != 0) {
```

```
    x = a - 1;
```

```
} else {
```

```
    if (y <= 10 || z == 42) {
```

```
        y = y + 1;
```

```
    } else {
```

```
        x = 0;
```

```
    }
```

```
}
```